Moed aleph exam 5770 - my answers to Maple questions. Note there are many ways to do everything!

Section A, Question 1

> $restart;\ with(LinearAlgebra):$

> $M := \langle\langle 1, \exp(x), x\rangle | \langle \exp(x), 1, x^2\rangle | \langle x, x^2, 1\rangle\rangle;$

$$M := \begin{bmatrix} 1 & e^x & x \\ e^x & 1 & x^2 \\ x & x^2 & 1 \end{bmatrix} \quad (1)$$

> $int(Trace(M^2), x = 0..1)$

$$\frac{46}{15} + e^2 \quad (2)$$

Section A, Question 3

> $restart;$

```
> f :=proc(v)
      local i, s;
      s := 0;
      for i from 1 to nops(v) do
          if floor(v[i]/2) = v[i]/2 then s := s + v[i] end if
     end do:
     return(s);
   end proc
```

```
f:= proc(v)
    local i, s;
    s:= 0;
    for i to nops(v) do
        if floor(1 / 2 * v[i]) = 1 / 2 * v[i] then s:= s + v[i] end if
    end do;
    return s
end proc                                                          (3)
```

> $f([1, 2, 3, 4, 5, 6, 7, 8])$

$$20 \quad (4)$$

Section A, Question 6

> $restart;\ with(LinearAlgebra):$

> $M := \langle\langle t+2, t\rangle | \langle 1, t^2\rangle\rangle$

$$M := \begin{bmatrix} t+2 & 1 \\ t & t^2 \end{bmatrix} \quad (5)$$

> $v := Eigenvalues(M)$

$$v := \begin{bmatrix} \frac{1}{2} t^2 + \frac{1}{2} t + 1 + \frac{1}{2} \sqrt{t^4 - 2 t^3 - 3 t^2 + 8 t + 4} \\ \frac{1}{2} t^2 + \frac{1}{2} t + 1 - \frac{1}{2} \sqrt{t^4 - 2 t^3 - 3 t^2 + 8 t + 4} \end{bmatrix} \quad (6)$$

> $plot([v[1], v[2], t = 2..5])$



> $evalf(int(\text{sqrt}(diff(v[1], t)^2 + diff(v[2], t)^2), t = 2..5) , 20)$

$$20.433427412453299435 \quad (7)$$

Section B, Question 3 - notes added after each line

> $p := \cos(3\cdot x) + 8\cdot\sin(3\cdot x) + 4\cdot\sin(2\cdot x) + \cos(x) + 7\cdot\sin(x)$

$$p := \cos(3\,x) + 8\,\sin(3\,x) + 4\,\sin(2\,x) + \cos(x) + 7\,\sin(x) \quad (8)$$

Assigning p to be the trigonometric polynomial we want to look at

> $p2 := simplify(expand(p))$

$$p2 := 4\,\cos(x)^3 - 2\,\cos(x) + 32\,\sin(x)\,\cos(x)^2 - \sin(x) + 8\,\sin(x)\,\cos(x) \quad (9)$$

This takes things like cos(3x) and writes them in terms of sin(x), cos(x) (cos(3x) = 4 cos(x)^3 - 3 cos(x) etc). There are many ways to do this, for example cos(2x) = 2 cos (x)^2 - 1 = cos(x)^2-sin(x)^2 = 1 - 2 sin(x)^2. The rule Maple follows is that whenever it sees sin(x)^2 it replaces it by 1-cos(x)^2. So the answer here is linear in sin(x). This is good as it will let us solve for sin(x) in terms of cos(x). This can be used to solve many similar problems.

> $p3 := subs(\sin(x) = s, \cos(x) = c, p2)$

$$p3 := 4\,c^3 - 2\,c + 32\,s\,c^2 - s + 8\,s\,c \quad \textbf{(10)}$$

Write "c" instead of cos(x) and "s" instead of sin(x)

> $mysin := solve(p3 = 0, s)$

$$mysin := -\frac{2\,c\,(2\,c^2 - 1)}{32\,c^2 - 1 + 8\,c} \quad \textbf{(11)}$$

Solve to find s in terms of c, calling the answer "mysin".

> $myeq := simplify(1 - c^2 - mysin^2)$

$$myeq := -\frac{-1040\,c^4 - 528\,c^3 - 1 + 16\,c + 1040\,c^6 + 512\,c^5 + 5\,c^2}{(32\,c^2 - 1 + 8\,c)^2} \quad \textbf{(12)}$$

So how does it help that we have found sin(x) in terms of cos(x) ? sin(x) and cos(x) must satisfy the equation sin(x)^2 + cos(x)^2 = 1. Substituting "mysin" into this gives a polynomial equation for c=cos(x). simplify just makes it all look nice.

> $numer(myeq)$

$$1040\,c^4 + 528\,c^3 + 1 - 16\,c - 1040\,c^6 - 512\,c^5 - 5\,c^2 \quad \textbf{(13)}$$

For myeq to vanish the numerator must vanish

> $cs := [fsolve(numer(myeq) = 0, c)]$

$$cs := [-0.9962842779, -0.3802247078, -0.3003437886, 0.07978347358, 0.1060665800, 0.9986950283] \quad \textbf{(14)}$$

Find (numerically) the roots of myeq, make them into a list, call them cs.

> $ss := [0\$6]$; **for** i **from** 1 **to** $nops(cs)$ **do** $ss[i] := subs(c = cs[i], mysin)$ **end do**: ss;

$$ss := [0, 0, 0, 0, 0, 0]$$

$$[0.08612570768, -0.9248941400, 0.9538310166, -0.9968122176, 0.9943590330, -0.05107093454] \quad \textbf{(15)}$$

For each of the cs (which are values of cos(x)) find the associated value of sin(x), call them ss. The first command here makes a list ss with 6 0's in it. The for loop puts the correct values in ss, by taking the corresponding value of cs and putting it in the formula mysin.

> $xs := [0\$6]$; **for** i **from** 1 **to** $nops(cs)$ **do** $xs[i] := \arctan(ss[i], cs[i])$ **end do**: xs;

$$xs := [0, 0, 0, 0, 0, 0]$$

$$[3.055360114, -1.960835567, 1.875849390, -1.490927968, 1.464529857, -0.05109316152] \quad \textbf{(16)}$$

We have the values of cos and the values of sin. To get the values of x apply the command arctan(a,b) ... this finds the angle with sin equal to a and cos equal to b. (More generally it finds the angle theta such that a = r sin(theta), b=r cos(theta) for some positive r). Again here the first command makes a list xs with 6 0's in, the loop puts the right value in the lists. The last command prints the answer.

> $plot(p, x = -\mathrm{Pi}..\mathrm{Pi})$



Make a graph of p. See 6 roots at the values of x given in xs.
Can you use it for other equations? In general you can try it for any equation of the form f(sin(x),cos(x),sin(2x),cos(2x), sin(3x),cos(3x),...)=0 where f is polynomial in its arguments. It does not work all the time, for example if s simply does not appear in p3 then you should just directly solve p3 - but this can have problems like complex roots and roots that are not in [-1,1]. The only thing that really needs changing is that the "6" in the above should be replaced by nops(cs).